Q:

Is it possible to ignore the storeFile when building a debug version of the application?

The crux of the issue is that not everyone working on this application has access to the storefile, and has to comment out these lines as followed in order to make a gradle sync:
signingConfigs {
    release {
//        storeFile file('.../android_keystore.keystore')
//        storePassword RELEASE_STORE_PASSWORD
//        keyAlias RELEASE_KEY_ALIAS
//        keyPassword RELEASE_KEY_PASSWORD
    }
}

In our build types, we defined that there is no signingconfig in debug:
buildTypes{
     release {
          ...
     }
     debug {
          singingConfig null
          ...
     }
}

The problem is that Gradle Syncs are agnostic to the build type, therefore it checks for the signing configs (storePassword, keyAlias, keyPassword) every time unless I comment those lines out.
Is there a more automated way to ignore these lines?

A:

You can use something like this:
android {

    signingConfigs {
        release
    }

    buildTypes {
            release {
                signingConfig signingConfigs.release
            }
    }
}

def Properties props = new Properties()
def propFile = new File('signing.properties')
if (propFile.canRead()){
    props.load(new FileInputStream(propFile))

    if (props!=null && props.containsKey('STORE_FILE') && props.containsKey('STORE_PASSWORD') &&
            props.containsKey('KEY_ALIAS') && props.containsKey('KEY_PASSWORD')) {
        android.signingConfigs.release.storeFile = file(props['STORE_FILE'])
        android.signingConfigs.release.storePassword = props['STORE_PASSWORD']
        android.signingConfigs.release.keyAlias = props['KEY_ALIAS']
        android.signingConfigs.release.keyPassword = props['KEY_PASSWORD']
    } else {
        println 'signing.properties found but some entries are missing'
        android.buildTypes.release.signingConfig = null
    }
}else {
    println 'signing.properties not found'
    android.buildTypes.release.signingConfig = null
}

where signing.properties is:
STORE_FILE=/path/to/your.keystore
STORE_PASSWORD=yourkeystorepass
KEY_ALIAS=projectkeyalias
KEY_PASSWORD=keyaliaspassword

A:

Marking @Gabriele's answer as the answer because it's far more complete and more correct, but wanted post an update about my final (easy and simple) solution, which I wouldn't have figured out without his answer;
All I had to do was check if the file existed. I didn't realize I could call methods and use if statements in the Gradle file;
signingConfigs {
     release {
          storeFile file('.../android_keystore.keystore')
          if (storeFile.exists()) {
               storePassword RELEASE_STORE_PASSWORD
               keyAlias RELEASE_KEY_ALIAS
               keyPassword RELEASE_KEY_PASSWORD
          }
}

